
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 7.1.1.4.2.1 which is part of the NR5GC test suite in the ‘iwd-TTCN3-B2019-12_D19wk23’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.1.4.2.1
Test Group:
\NR5GC\7_1_1\
ATS Version:
iwd-TTCN3-B2019-12_D19wk23
System Simulator used:
Keysight 5G Protocol Conformance Toolset
UE used:
HiSilicon Balong 5000
Verification Status:
PASS


4. Corrections required for NR5GC TC 7.1.4.2.1 on wk23 ATS
4.1 Change 1

	Function name
	TC_7_1_1_4_2_1_NR5GC

	Reason for change
	Succeful test case execution can sometimes take longer than the configured guard timer of 1200s

	Summary of change
	Extend guard timer

	TTCN module
	MAC_NR5GC

	MCC160 Comment
	


Before 

   testcase TC_7_1_1_4_2_1_NR5GC() runs on MTC_NR5GC system SYSTEM_NR5GC {

    // @purpose

    //   DL-SCH Transport Block Size selection / DCI format 1_0

    var NR5GC_PTC        v_NR5GC      := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(1200);

    v_NR5GC := NR5GC_PTC.create alive;

    f_MTC_ConnectPTCs_NR5GC(system, v_NR5GC, v_IMS1, v_IMS2);

    v_NR5GC.start(f_TC_7_1_1_4_1_1_NR5GC());

    t_GuardTimer.start;

    f_MTC_MainLoop_NR5GC(t_GuardTimer);

  }
After change

 testcase TC_7_1_1_4_2_1_NR5GC() runs on MTC_NR5GC system SYSTEM_NR5GC {

    // @purpose

    //   DL-SCH Transport Block Size selection / DCI format 1_0

    var NR5GC_PTC        v_NR5GC      := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(6000);

    v_NR5GC := NR5GC_PTC.create alive;

    f_MTC_ConnectPTCs_NR5GC(system, v_NR5GC, v_IMS1, v_IMS2);

    v_NR5GC.start(f_TC_7_1_1_4_1_1_NR5GC());

    t_GuardTimer.start;

    f_MTC_MainLoop_NR5GC(t_GuardTimer);

  }
4.2 Change 2

	Function name
	f_TC_7_1_1_4_2_1_NR5GC()

	Reason for change
	1. K2 value should be 2 for FR1
2. SS should be configured with MAC, RLC and PDCP in normal mode. No SDAP is configured because the test case body is dealing with PDCP SDUs.
3. DRB ID to be modified does not need to be passed into f_NR5GC_RRC_ConnectedState3N_Def() because p_DRBToAddModList is not being passed

	Summary of change
	1. Correct K2 value for FR1
2. Pass SS DRB configuration to f_NR5GC_RRC_ConnectedState3N_Def()
3. Omit DRB ID to be modified in call to f_NR5GC_RRC_ConnectedState3N_Def()

	TTCN module
	MAC_NR5GC

	MCC160 Comment
	


Before change

     function f_TC_7_1_1_4_2_1_NR5GC() runs on NR5GC_PTC

  { // UL-SCH Transport Block Size selection / DCI format 0_0 / Transform precoding disabled

    var DRB_Identity v_DRBId ;

    var template (value) NR_TBS_CalculationParameters_Type v_CalculationParameters:= cs_NR_TBS_CalculationParameters_711421;

    var integer v_K2 := 4; // FR2

    //Init Cell parameters

    f_NR5GC_Init(NR_1);

    if ( f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_K2 := 7};
    f_NR_Init_7_1_1_4_2_x(nr_Cell1, formats0_0_And_1_0, v_K2, omit, omit);

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    // Get NR capabilities sent during EUTRA preamble

    f_NR5GC_Preamble (nr_Cell1, STATE_IDLE_1A, TESTModeA_ON);

        //Retrieve DRB Id of the first PDU session

    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    //Configure on SS the DRB in No MAC header manipulation in DL and RLC and PDCP in transparent mode

    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,

                                       TEST_LOOPModeA_ON,

                                       cs_UE_TestLoopModeA_NR_LB_Setup_NoScaling,

                                       -,  //p_IpPduDelayTime

                                       {v_DRBId},  //DRBId_ToBeModified
                                       -,

                                       -,  //MAC_CellGroupConfig

                                       -,

                                       -
                                       );

    f_NR_TestBody_Set(true);

    f_TC_7_1_1_4_2_x_DCI0_0_NR_TestBody(v_DRBId,v_CalculationParameters);

    f_NR_TestBody_Set(false);

     //Postamble

     f_NR_ULGrantConfiguration_Start(nr_Cell1); // Def UL Grant Config

    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  };
After change

function f_TC_7_1_1_4_2_1_NR5GC() runs on NR5GC_PTC

  { // UL-SCH Transport Block Size selection / DCI format 0_0 / Transform precoding disabled

    var DRB_Identity v_DRBId ;

    var template (value) NR_TBS_CalculationParameters_Type v_CalculationParameters:= cs_NR_TBS_CalculationParameters_711421;

    var integer v_K2 := 4; // FR2

    var template (value) NR_RadioBearerList_Type v_SS_Drb_Config;
    //Init Cell parameters

    f_NR5GC_Init(NR_1); 

    if ( f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_K2 := 2};
    f_NR_Init_7_1_1_4_2_x(nr_Cell1, formats0_0_And_1_0, v_K2, omit, omit);

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    // Get NR capabilities sent during EUTRA preamble

    f_NR5GC_Preamble (nr_Cell1, STATE_IDLE_1A, TESTModeA_ON);

        //Retrieve DRB Id of the first PDU session

    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();

    v_SS_Drb_Config := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1, v_DRBId, -, -, -, cs_SDAP_Configuration_None)};
    //Configure on SS the DRB in No MAC header manipulation in DL and RLC and PDCP in transparent mode

    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,

                                       TEST_LOOPModeA_ON,

                                       cs_UE_TestLoopModeA_NR_LB_Setup_NoScaling,

                                       -,  //p_IpPduDelayTime

                                       -,
                                       -,

                                       -,  //MAC_CellGroupConfig

                                       -,

                                       v_SS_Drb_Config
                                       );

    f_NR_TestBody_Set(true);

    f_TC_7_1_1_4_2_x_DCI0_0_NR_TestBody(v_DRBId,v_CalculationParameters);

    f_NR_TestBody_Set(false);

     //Postamble

     f_NR_ULGrantConfiguration_Start(nr_Cell1); // Def UL Grant Config

    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);

    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

  };
Change 3
	Function name
	fl_NR_GetPdcpSduList_1TB

	Reason for change
	According to Table 7.1.1.4.2.1.3.2-2, SDU size should be TBS minus 344 when 36273 <= p_TBSize <= 48344.

	Summary of change
	Correct the calculation of PDCP SDU size when 36273 <= p_TBSize <= 48344.

	TTCN module
	MAC_TC_Common_NR

	MCC160 Comment
	


Before 

function fl_NR_GetPdcpSduList_1TB(integer p_TBSize) return NR_PDCP_SDUList_Type

  {

    var integer v_N;

    var integer v_PdcpSduSize;

    if (p_TBSize <= 12128) {

      v_N := 1;

      v_PdcpSduSize := (p_TBSize - 128)/8;

    }

    else if (p_TBSize <= 24200) {

      v_N := 2;

      v_PdcpSduSize := (p_TBSize - 200)/16;

    }

    else if (p_TBSize <= 36272) {

      v_N := 3;

      v_PdcpSduSize := (p_TBSize - 272)/24;

    }

    else if (p_TBSize <= 48344) {

      v_N := 4;

      v_PdcpSduSize := (p_TBSize - 244)/32;
    }

    else if (p_TBSize <= 60416) {

      v_N := 5;

      v_PdcpSduSize := (p_TBSize - 416)/40;

    }

    …..
    return f_NR_Build_N_PDCP_SDUList(v_N, v_PdcpSduSize);

  }
After change

function fl_NR_GetPdcpSduList_1TB(integer p_TBSize) return NR_PDCP_SDUList_Type

  {

    var integer v_N;

    var integer v_PdcpSduSize;

    if (p_TBSize <= 12128) {

      v_N := 1;

      v_PdcpSduSize := (p_TBSize - 128)/8;

    }

    else if (p_TBSize <= 24200) {

      v_N := 2;

      v_PdcpSduSize := (p_TBSize - 200)/16;

    }

    else if (p_TBSize <= 36272) {

      v_N := 3;

      v_PdcpSduSize := (p_TBSize - 272)/24;

    }

    else if (p_TBSize <= 48344) {

      v_N := 4;

      v_PdcpSduSize := (p_TBSize - 344)/32;
    }

    else if (p_TBSize <= 60416) {

      v_N := 5;

      v_PdcpSduSize := (p_TBSize - 416)/40;

    }

    …..
    return f_NR_Build_N_PDCP_SDUList(v_N, v_PdcpSduSize);

  }
Change 4
	Function name
	f_NR_Init_7_1_1_4_2_x

	Reason for change
	As per 38.523 Table 7.1.1.4.2.0-1, the first entry in PUSCH-TimeDomainResourceAllocationList is different for FR1 and FR2. Template cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x is using the FR2 value.

	Summary of change
	Use a different template for PUSCH-TimeDomainResoureAllocationList for FR1 operation (See change 5)

	TTCN module
	MAC_TC_Common_NR

	MCC160 Comment
	


Before 

function f_NR_Init_7_1_1_4_2_x(NR_CellId_Type p_NR_CellId,

                                 NR_SearchSpace_SearchSpaceType_Type.ue_Specific.dci_Formats p_DciFormats,

                                 PUSCH_TimeDomainResourceAllocation.k2 p_K2,

                                 template (omit) PUSCH_Config.mcs_Table p_Mcs_Table,

                                 template (omit) PUSCH_Config.transformPrecoder p_TransformPrecoder) runs on NR_BASE_PTC

  {

    var BWP_Id v_BWP_Id := tsc_NR_BWP_Id;

    var template (value) NR_UplinkBWP_Type v_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(p_NR_CellId, v_BWP_Id);

    var template (value) NR_DownlinkBWP_Type v_DownlinkBWP := f_NR_CellInfo_GetDownlinkBWP(p_NR_CellId, v_BWP_Id);

    var integer v_SearchSpaceIndex := f_NR_DownlinkBWP_GetSearchSpace_Index(v_DownlinkBWP, tsc_NR_SearchSpaceId_USS);

    v_UplinkBWP.Dedicated.R15.pusch_Config.setup.pusch_TimeDomainAllocationList.setup := cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x(p_K2);
    v_UplinkBWP.Dedicated.R15.pusch_Config.setup.mcs_Table  := p_Mcs_Table; //qam64 if omit

    v_UplinkBWP.Dedicated.R15.pusch_Config.setup.transformPrecoder :=p_TransformPrecoder;

    if (match(enabled, p_TransformPrecoder)) {

      v_UplinkBWP.Dedicated.R15.pusch_Config.setup.mcs_TableTransformPrecoder := omit;

    }

    v_DownlinkBWP.Pdcch.SearchSpaceArray[v_SearchSpaceIndex].SearchSpaceConfigAtUE.R15.searchSpaceType.ue_Specific.dci_Formats := p_DciFormats;

    f_NR_CellInfo_SetUplinkBWP(p_NR_CellId, v_UplinkBWP, v_BWP_Id);

    f_NR_CellInfo_SetDownlinkBWP(p_NR_CellId, v_DownlinkBWP, v_BWP_Id);

  }
After change

function f_NR_Init_7_1_1_4_2_x(NR_CellId_Type p_NR_CellId,

                                 NR_SearchSpace_SearchSpaceType_Type.ue_Specific.dci_Formats p_DciFormats,

                                 PUSCH_TimeDomainResourceAllocation.k2 p_K2,

                                 template (omit) PUSCH_Config.mcs_Table p_Mcs_Table,

                                 template (omit) PUSCH_Config.transformPrecoder p_TransformPrecoder) runs on NR_BASE_PTC

  {

    var BWP_Id v_BWP_Id := tsc_NR_BWP_Id;

    var template (value) NR_UplinkBWP_Type v_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(p_NR_CellId, v_BWP_Id);

    var template (value) NR_DownlinkBWP_Type v_DownlinkBWP := f_NR_CellInfo_GetDownlinkBWP(p_NR_CellId, v_BWP_Id);

    var integer v_SearchSpaceIndex := f_NR_DownlinkBWP_GetSearchSpace_Index(v_DownlinkBWP, tsc_NR_SearchSpaceId_USS);

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1))

    {

      v_UplinkBWP.Dedicated.R15.pusch_Config.setup.pusch_TimeDomainAllocationList.setup := cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x_FR1(p_K2);

    } else

    {

      v_UplinkBWP.Dedicated.R15.pusch_Config.setup.pusch_TimeDomainAllocationList.setup := cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x(p_K2);

    }
    v_UplinkBWP.Dedicated.R15.pusch_Config.setup.mcs_Table  := p_Mcs_Table; //qam64 if omit

    v_UplinkBWP.Dedicated.R15.pusch_Config.setup.transformPrecoder :=p_TransformPrecoder;

    if (match(enabled, p_TransformPrecoder)) {

      v_UplinkBWP.Dedicated.R15.pusch_Config.setup.mcs_TableTransformPrecoder := omit;

    }

    v_DownlinkBWP.Pdcch.SearchSpaceArray[v_SearchSpaceIndex].SearchSpaceConfigAtUE.R15.searchSpaceType.ue_Specific.dci_Formats := p_DciFormats;

    f_NR_CellInfo_SetUplinkBWP(p_NR_CellId, v_UplinkBWP, v_BWP_Id);

    f_NR_CellInfo_SetDownlinkBWP(p_NR_CellId, v_DownlinkBWP, v_BWP_Id);

  }
Change 5
	Function name
	Template cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x_FR1

	Reason for change
	As per 38.523 Table 7.1.1.4.2.0-1, the first entry in PUSCH-TimeDomainResourceAllocationList is different for FR1 and FR2. Template cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x is using the FR2 value.

	Summary of change
	New template for PUSCH-TimeDomainResoureAllocationList for FR1 operation

	TTCN module
	MAC_TC_Common_NR

	MCC160 Comment
	


New Template
template (value) PUSCH_TimeDomainResourceAllocationList cs_NR_PUSCH_TimeDomainResourceAllocationList_71142x_FR1 (PUSCH_TimeDomainResourceAllocation.k2 p_K2):=

  { // Acc to 38.523-1 Table 7.1.1.4.2.0-1: PUSCH-TimeDomainResourceAllocationList

    cs_NR_PUSCH_TimeDomainResourceAllocation(p_K2, typeB, 52),  //Start symbol(S)=10, Length(L)=4

    cs_NR_PUSCH_TimeDomainResourceAllocation(p_K2, typeB, 27)   //Start symbol(S)=0, Length(L)=14

  };
Change 6
	Function name
	f_TC_7_1_1_4_2_x_DCI0_0_NR_TestBody

	Reason for change
	1. Default ULgrant allocation should be stopped before entering the main test case loop. 
2. When MCS index is 28, the effective coding rate for PUSCH will up to 0.96 if v_N_PRB is 1 and has only 4 time domain symbols, which is undecodable for SS.
3. Only two UL time domain resources are configured in this test case. Need to change the range of Time Domain Resource Allocation from 0 to 1.
4. When calculating TBSize, the number of symbols is needed instead of the index of Time Domain Resource Allocation.
5. K2 is changed in this test case, need to use new scheduling parameter when get UL timing info.

	Summary of change
	1. Call function f_NR_ULGrantConfiguration_Stop to stop the UL grant before the loop.
2. Looping MCS index from 0 to 27.

Note : An associated prose CR on 38.523-1 will be raised at RAN5#85
3. Looping Time Domain Resource Allocation from 0 to 1.
4. Use a new variable v_NoOfSymbols to record the number of time domain symbols for the current time domain resource allocation and use it while calculating TBSize.
5. Call the new function provided in Change 8 to calculate UL timing info.

	TTCN module
	MAC_TC_Common_NR

	MCC160 Comment
	


Before 

function f_TC_7_1_1_4_2_x_DCI0_0_NR_TestBody(DRB_Identity p_NR_DRB_Id,

                                               template (value) NR_TBS_CalculationParameters_Type p_CalculationParameters) runs on NR_BASE_PTC

  { /* L2/MAC/7.1

     * 7.1.1.4.2.x: UL-SCH Transport Block Size selection

     */

    var integer v_TBSize:=0;    // transport block size in bits

    var integer v_I_MCS:=0;     // Modulation and Coding Scehme Index

    var integer v_N_PRB:=1;     // Number of physical resource Blocks

    var integer v_Max_nPRB := f_NR_CellInfo_GetCarrierBandwidth_DL(nr_Cell1);     // to store the max allowed PRB resources that can be allocated

    var integer v_TDRA ; // Time Domain resource allocation

    var NR_PDCP_SDUList_Type v_NR_PDCP_SDUList := {''O};

    var SubFrameTiming_Type v_Timing;

    // Configure SS for DL grant for Max TB size

    f_NR_DLGrantConfiguration(nr_Cell1,

                              cs_TimingInfo_Now,

                              resourceAllocationType1,

                              cs_NR_FreqDomainResourceAssignmentDL_Explicit(0, v_Max_nPRB),

                              int2bit(0, 4),

                              cs_NR_TransportBlockRetransmissionList_Def(28),

                              -, -,

                              cs_NR_DciFormat_1_0_Params);

    //Steps 1 to 5 are repeated for values of nPRB  from 1 to v_Max_nPRB and iMCS from 0 to 28, time domain resource allocation v_TDRA 0..8

    //f_NR_ConfigureULGrant_MaxTBSupported();

    for (v_N_PRB := 1; v_N_PRB <= v_Max_nPRB; v_N_PRB := v_N_PRB + 1) {  //for nPRB 1 to v_Max_nPRB

      for (v_I_MCS := 0; v_I_MCS <= 28 ; v_I_MCS := v_I_MCS + 1) {   //for iMCS 0 to 28
        for (v_TDRA := 0; v_TDRA <= 13 ; v_TDRA := v_TDRA + 1) {   //for Time Domain Resource Allocation 0 to 13
          //SS looks up iTBS in table 7.1.7.1-1 in TS 36.213 based on the value of iMCS

          //v_I_TBS := f_NR_Get_iTBS_For_iMCS(v_I_MCS);

          //The SS calculates or looks up TBS in TS 38.214 [15] based on the value of S, L, and  nPRB.

          v_TBSize := f_NR_TBS_Calculation(p_CalculationParameters, v_TDRA, v_N_PRB, v_I_MCS);
          if (v_TBSize < 136) {            //this results in zero number of PDCP SDUs

            continue;

          }  // skip this nprb, Imcs, TDRA combination.

          // Steps 2 to 5 are performed if  TBsize is More than minimum

          //@siclog "Step 2" siclog@

          // SS creates one or more PDCP SDUs, depending on TBsize, in accordance with Table  7.1.1.4.1.1.3.2-2

          //Configure SS for DCI Format x with RA type y and a Resource block assignment (RBA) correspondent

          //to nPRB as specified in 7.1.6.1 in TS 36.213 and Modulation and coding scheme iMCS

          //@siclog "Step 3" siclog@

          // SS sends MAC PDU (NxPDCP SDUs)

          v_NR_PDCP_SDUList := fl_NR_GetPdcpSduList_1TB(v_TBSize);   //calculates v_N, v_PDCPsize and returns f_NR_Build_N_PDCP_SDUList

          v_Timing := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300); //300 ms in future

          DRB.send(cas_NR_DRB_COMMON_REQ_PDCP_SDUList(nr_Cell1,

                                                      p_NR_DRB_Id,

                                                      //crs_RlcBearerRouting_NR(nr_Cell1),

                                                      cs_TimingInfo_NR(v_Timing),

                                                      v_NR_PDCP_SDUList));

          //@siclog "Step 4" siclog@

          v_Timing := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_Timing, 60);
          f_NR_ULGrantConfiguration_ULTBS(nr_Cell1,

                                          cs_TimingInfo_NR(v_Timing),

                                          v_I_MCS,

                                          v_N_PRB,

                                          int2bit(v_TDRA, 4));

          //@siclog "Step 5" siclog@

          // SS receives NxPDCP SDUs

          f_NR_Rx_N_PDCP_SDUs_InOneTTI(v_NR_PDCP_SDUList, p_NR_DRB_Id);

        }//End for TDRA 0 to 13

      }//End for iMCS 0 to 28

    }//End for nPRB 1 to 110

  }//f_NR_7_1_1_4_2_1
After change

function f_TC_7_1_1_4_2_x_DCI0_0_NR_TestBody(DRB_Identity p_NR_DRB_Id,

                                               template (value) NR_TBS_CalculationParameters_Type p_CalculationParameters) runs on NR_BASE_PTC

  { /* L2/MAC/7.1

     * 7.1.1.4.2.x: UL-SCH Transport Block Size selection

     */

    var integer v_TBSize:=0;    // transport block size in bits

    var integer v_I_MCS:=0;     // Modulation and Coding Scehme Index

    var integer v_N_PRB:=1;     // Number of physical resource Blocks

    var integer v_Max_nPRB := f_NR_CellInfo_GetCarrierBandwidth_DL(nr_Cell1);     // to store the max allowed PRB resources that can be allocated

    var integer v_TDRA ; // Time Domain resource allocation

    var integer v_NoOfSymbols ;//Number of symbols in slot
    var NR_PDCP_SDUList_Type v_NR_PDCP_SDUList := {''O};

    var SubFrameTiming_Type v_Timing;

    // Configure SS for DL grant for Max TB size

    f_NR_DLGrantConfiguration(nr_Cell1,

                              cs_TimingInfo_Now,

                              resourceAllocationType1,

                              cs_NR_FreqDomainResourceAssignmentDL_Explicit(0, v_Max_nPRB),

                              int2bit(0, 4),

                              cs_NR_TransportBlockRetransmissionList_Def(28),

                              -, -,

                              cs_NR_DciFormat_1_0_Params);

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);
    //Steps 1 to 5 are repeated for values of nPRB  from 1 to v_Max_nPRB and iMCS from 0 to 28, time domain resource allocation v_TDRA 0..8

    //f_NR_ConfigureULGrant_MaxTBSupported();

    for (v_N_PRB := 1; v_N_PRB <= v_Max_nPRB; v_N_PRB := v_N_PRB + 1) {  //for nPRB 1 to v_Max_nPRB

      for (v_I_MCS := 0; v_I_MCS <= 27 ; v_I_MCS := v_I_MCS + 1) {   //for iMCS 0 to 27
        for (v_TDRA := 0; v_TDRA <= 1 ; v_TDRA := v_TDRA + 1) {   //for Time Domain Resource Allocation 0 to 1
          //SS looks up iTBS in table 7.1.7.1-1 in TS 36.213 based on the value of iMCS

          //v_I_TBS := f_NR_Get_iTBS_For_iMCS(v_I_MCS);

          //The SS calculates or looks up TBS in TS 38.214 [15] based on the value of S, L, and  nPRB.

          if(v_TDRA == 0)

          {

            v_NoOfSymbols := 4;

          }

          else{

            v_NoOfSymbols := 14

          }

          v_TBSize := f_NR_TBS_Calculation(p_CalculationParameters, v_NoOfSymbols, v_N_PRB, v_I_MCS);
          if (v_TBSize < 136) {            //this results in zero number of PDCP SDUs

            continue;

          }  // skip this nprb, Imcs, TDRA combination.

          // Steps 2 to 5 are performed if  TBsize is More than minimum

          //@siclog "Step 2" siclog@

          // SS creates one or more PDCP SDUs, depending on TBsize, in accordance with Table  7.1.1.4.1.1.3.2-2

          //Configure SS for DCI Format x with RA type y and a Resource block assignment (RBA) correspondent

          //to nPRB as specified in 7.1.6.1 in TS 36.213 and Modulation and coding scheme iMCS

          //@siclog "Step 3" siclog@

          // SS sends MAC PDU (NxPDCP SDUs)

          v_NR_PDCP_SDUList := fl_NR_GetPdcpSduList_1TB(v_TBSize);   //calculates v_N, v_PDCPsize and returns f_NR_Build_N_PDCP_SDUList

          v_Timing := f_NR_GetNextSendOccasion_DL(nr_Cell1, 300); //300 ms in future

          DRB.send(cas_NR_DRB_COMMON_REQ_PDCP_SDUList(nr_Cell1,

                                                      p_NR_DRB_Id,

                                                      //crs_RlcBearerRouting_NR(nr_Cell1),

                                                      cs_TimingInfo_NR(v_Timing),

                                                      v_NR_PDCP_SDUList));

          //@siclog "Step 4" siclog@ 

          v_Timing := f_NR_GetNextSendOccasion_ULafterDL_71142x(nr_Cell1, v_Timing, 60);
          f_NR_ULGrantConfiguration_ULTBS(nr_Cell1,

                                          cs_TimingInfo_NR(v_Timing),

                                          v_I_MCS,

                                          v_N_PRB,

                                          int2bit(v_TDRA, 4));

          //@siclog "Step 5" siclog@

          // SS receives NxPDCP SDUs

          f_NR_Rx_N_PDCP_SDUs_InOneTTI(v_NR_PDCP_SDUList, p_NR_DRB_Id);

        }//End for TDRA 0 to 13

      }//End for iMCS 0 to 28

    }//End for nPRB 1 to 110

  }//f_NR_7_1_1_4_2_1
Change 7
	Function name
	fl_NR_SchedulingParameters_UL_71142x

	Reason for change
	Since K2 is changed in this test case, need to provide new scheduling parameters based on the new K2. 

	Summary of change
	Add a new function to provide the new scheduling parameters. This will be used in Change 8.

	TTCN module
	NR_Timing

	MCC160 Comment
	


New Function

function fl_NR_SchedulingParameters_UL_71142x(NR_CellId_Type p_CellId) runs on NR_BASE_PTC return NR_SchedulingParameters_Type

  {

    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_CellId);

    var UInt_Type v_SlotsPerSubframe := f_NR_CellInfo_GetSCS_SlotsNumberPerSubframe(v_SCS);

    var NR_SchedulingParameters_Type v_SchedulingParameters;

    v_SchedulingParameters := {

      ValidSubframeIndex := 4,                    /* => with half-frame periodicity subframes 4 and 9 are valid for UL transmissions */

      SubframePeriodicity := 5,                   /* half-frame */

      ValidSlotNumber := 0,  /* first slot in a subframe */

      SlotOffsetPdcchPxsch := 2

    }

    return v_SchedulingParameters;

  }
Change 8
	Function name
	f_NR_GetNextSendOccasion_ULafterDL_71142x

	Reason for change
	Since K2 is changed in this test case, need to provide a new function to calculate the correct UL timing info.

	Summary of change
	Add a new function to calculate the correct UL timing info based on new K2 (see change 7)

	TTCN module
	NR_Timing

	MCC160 Comment
	


New Function

function f_NR_GetNextSendOccasion_ULafterDL_71142x(NR_CellId_Type p_CellId,

                                              SubFrameTiming_Type p_T_DlAssignment,

                                              UInt_Type p_MilliSeconds) runs on NR_BASE_PTC return SubFrameTiming_Type

  {

    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_CellId);

    var UInt_Type v_SlotsPerSubframe := f_NR_CellInfo_GetSCS_SlotsNumberPerSubframe(v_SCS);

    var NR_SchedulingParameters_Type v_SchedulingParametersDL := fl_NR_SchedulingParameters_DL();

    var NR_SchedulingParameters_Type v_SchedulingParametersUL := fl_NR_SchedulingParameters_UL_71142x(p_CellId);

    var UInt_Type v_SlotOffsetK0 := v_SchedulingParametersDL.SlotOffsetPdcchPxsch;

    var UInt_Type v_SlotNumber := f_NR_SubFrameTiming_GetSlotNumber(p_T_DlAssignment);

    var UInt_Type v_SubframeOffsetForPdschTransmission := ((p_T_DlAssignment.Subframe.Number * v_SlotsPerSubframe) + v_SlotNumber + v_SlotOffsetK0) / v_SlotsPerSubframe;

    return fl_NR_NextPdcchOcassion(p_CellId, p_T_DlAssignment.HSFN, p_T_DlAssignment.SFN, v_SubframeOffsetForPdschTransmission + p_MilliSeconds, v_SchedulingParametersUL);

  }
Change 9
	Function name
	NR_DciUlInfo_Type cds_NR_DciUlInfo_ModifyGrant

	Reason for change
	This function doesn’t take p_TimeDomainResourceAssignment into consideration when calling cs_NR_DciFormat_0_X_ResourceAssignment().

	Summary of change
	Pass p_TimeDomainResourceAssignment as one of the parameters when calling cs_NR_DciFormat_0_X_ResourceAssignment().

	TTCN module
	NR_DCI_Configuration

	MCC160 Comment
	


Before 

template (value) NR_DciUlInfo_Type cds_NR_DciUlInfo_ModifyGrant(template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul,

                                                                  template (omit) NR_DciCommon_TimeDomainResourceAssignment_Type p_TimeDomainResourceAssignment := omit,

                                                                  template (value) NR_ImcsValue_Type p_ImcsValue)

    modifies cs_NR_DciUlInfo_AllOmit :=

  { /* @status    APPROVED (ENDC) */

    ResoureAssignment := cs_NR_DciFormat_0_X_ResourceAssignment(-, p_FreqDomainSchedul, -, cs_NR_TransportBlockRetransmissionList_Def(p_ImcsValue))
  };
After change

template (value) NR_DciUlInfo_Type cds_NR_DciUlInfo_ModifyGrant(template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul,

                                                                  template (omit) NR_DciCommon_TimeDomainResourceAssignment_Type p_TimeDomainResourceAssignment := omit,

                                                                  template (value) NR_ImcsValue_Type p_ImcsValue)

    modifies cs_NR_DciUlInfo_AllOmit :=

  { /* @status    APPROVED (ENDC) */

    ResoureAssignment := cs_NR_DciFormat_0_X_ResourceAssignment(-, p_FreqDomainSchedul, p_TimeDomainResourceAssignment, cs_NR_TransportBlockRetransmissionList_Def(p_ImcsValue))

  };
5. Execution Log Files

5.1 HiSilicon Balong 5000
The HiSilicon Balong 5000UE passed this test case on Keysight 5G Protocol Conformance Toolset Solution on NR5GC band n41.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_7_1_1_4_2_1_Log.html
· 

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

